
Key to Midterm Exam S4
Computer Architecture

Duration: 1 hr 30 min

Write answers only on the answer sheet.

Exercise     1  (4 points)
Complete the table shown on the  answer sheet. Write down the new values of the registers (except the

PC) and memory that are modified by the instructions. Use the hexadecimal representation. Memory

and registers are reset to their initial values for each instruction.

Initial values: D0 = $FFFF0015  A0 = $00005000  PC = $00006000

D1 = $12340004  A1 = $00005008

D2 = $FFFFFFE1  A2 = $00005010

$005000  54 AF 18 B9 E7 21 48 C0

$005008  C9 10 11 C8 D4 36 1F 88

$005010  13 79 01 80 42 1A 2D 49

Exercise     2  (3 points)
Complete the table shown on the answer sheet. Determine the missing number for each addition in order

to match the given flags (use the hexadecimal representation). If multiple answers are possible, choose

the smallest one.

Exercise     3  (4 points)
Let us consider the following program. Complete the table shown on the answer sheet.

Main        move.l  #$85A51000,d7

next1       moveq.l #1,d1
            cmpi.w  #$80,d7
            blt     next2
            moveq.l #2,d1

next2       move.l  d7,d2
            ror.l   #4,d2
            swap    d2
            rol.w   #8,d2
            rol.b   #4,d2

next3       clr.l   d3
            move.l  d7,d0
loop3       addq.l  #1,d3
            subq.w  #2,d0
            bne     loop3

next4       clr.l   d4
            move.l  d7,d0
loop4       addq.l  #1,d4
            dbra    d0,loop4      ; DBRA = DBF

Exercise     4  (9 points)
All questions in this exercise are independent. Except for the output registers, none of the data or ad-

dress registers must be modified when the subroutine returns. Be careful.   All the   subroutines must  

contain   15   lines of instructions at the most.  

Structure of a bitmap:

Field
Size

(bits)
Encoding Description

WIDTH 16 Unsigned integer Width of the bitmap in pixels

HEIGHT 16 Unsigned integer Height of the bitmap in pixels

MATRIX Variable Bitmap

Dot matrix of the bitmap.

If a bit is 0, the displayed pixel is black.

If a bit is 1, the displayed pixel is white.

Structure of a sprite:

Field
Size

(bits)
Encoding Description

STATE 16 Unsigned integer
Current display state of the sprite

Only two possible values: HIDE = 0 or SHOW = 1

X 16 Signed integer Abscissa of the sprite

Y 16 Signed integer Ordinate of the sprite

BITMAP1 32 Unsigned integer Address of the first bitmap

BITMAP2 32 Unsigned integer Address of the second bitmap

We assume that the size of the bitmap 1 is always equal to that of the bitmap 2.

Constants that are already defined:

VIDEO_START         equ     $ffb500         ; Starting address of the video memory
VIDEO_SIZE          equ     (480*320/8)     ; Size in bytes of the video memory

WIDTH               equ     0
HEIGHT              equ     2
MATRIX              equ     4

STATE               equ     0
X                   equ     2
Y                   equ     4
BITMAP1             equ     6
BITMAP2             equ     10

HIDE                equ     0
SHOW                equ     1

1. Write the FillScreen subroutine that fills the video memory with a 32-bit integer.

Input: D0.L = A 32-bit integer used to fill the video memory.

2. Write the  GetRectangle subroutine that returns the coordinates of the rectangle that marks out the

boundaries of a sprite.

Input: A0.L = Address of the sprite.

Outputs: D1.W  = Abscissa of the top left corner of the sprite.

D2.W  = Ordinate of the top left corner of the sprite.

D3.W  = Abscissa of the bottom right corner of the sprite.

D4.W  = Ordinate of the bottom right corner of the sprite.

3. Write the  MoveSprite subroutine that moves a sprite in a relative way. If the new position of the

sprite is off the screen, the sprite must remain still (the new position will be ignored).

Inputs: A1.L = Address of a sprite.

D1.W = Relative horizontal displacement in pixels (16-bit signed integers).

D2.W = Relative vertical displacement in pixels (16-bit signed integers).

Outputs: D0.L returns false (0) if the sprite has not moved (its new position was out of the screen).

D0.L returns true (1) if the sprite has moved.

To  know if a sprite is out of the screen, you can call the IsOutOfScreen subroutine. We will assume

that this subroutine has already been written (you do not have to write it).

Inputs: A0.L = Address of a bitmap.

D1.W = Abscissa of the bitmap in pixels (16-bit signed integer).

D2.W = Ordinate of the bitmap in pixels (16-bit signed integer).

Outputs: Z returns false (0) if the bitmap is not out of the screen.

Z returns true (1) if the bitmap is out of the screen.

Last name: ............................................. First name: ........................................... Group: ............................ 

ANSWER SHEET TO BE HANDED IN

Exercise     1  

Instruction Memory Register

Example $005000  54 AF 00 40 E7 21 48 C0
A0 = $00005004

A1 = $0000500C

Example $005008  C9 10 11 C8 D4 36 FF 88 No change

MOVE.L #4507,-(A1) $005000  54 AF 18 B9 00 00 11 9B A1 = $00005004

MOVE.B $5009,-6(A1) $005000  54 AF 10 B9 E7 21 48 C0 No change

MOVE.W 8(A1),-37(A2,D0.W) $005000  13 79 18 B9 E7 21 48 C0 No change

MOVE.L -4(A2),$21(A0,D2.L) $005000  54 AF D4 36 1F 88 48 C0 No change

Exercise     2  

Operation
Size

(bits)

Missing Number

(hexadecimal)
N Z V C

$80 + $? 8 $00 1 0 0 0

$8000 + $? 16 $8000 0 1 1 1

$80000000 + $? 32 $80000001 0 0 1 1

Exercise     3  

Values of registers after the execution of the program.

Use the 32-bit hexadecimal representation.

D1 = $00000002 D3 = $00000800

D2 = $51005A80 D4 = $00001001

Exercise     4  

FillScreen          movem.l d7/a0,-(a7)

                    lea     VIDEO_START,a0
                    move.w  #VIDEO_SIZE/4-1,d7
                    
\loop               move.l  d0,(a0)+
                    dbra    d7,\loop
                    
                    movem.l (a7)+,d7/a0
                    rts

GetRectangle        move.l  a0,-(a7)

                    move.w  X(a0),d1
                    move.w  Y(a0),d2

                    movea.l BITMAP1(a0),a0

                    move.w  WIDTH(a0),d3
                    add.w   d1,d3
                    subq.w  #1,d3

                    move.w  HEIGHT(a0),d4
                    add.w   d2,d4
                    subq.w  #1,d4

                    movea.l (a7)+,a0
                    rts

MoveSprite          movem.l d1/d2/a0,-(a7)

                    add.w   X(a1),d1
                    add.w   Y(a1),d2

                    movea.l BITMAP1(a1),a0

                    jsr     IsOutOfScreen
                    beq     \false

                    move.w  d1,X(a1)
                    move.w  d2,Y(a1)

\true               moveq.l #1,d0
                    bra     \quit

\false              moveq.l #0,d0
\quit               movem.l (a7)+,d1/d2/a0
                    rts

